

```
> p := 11;
p := 11 (1)
```

```
> d := 5; # d := 10;
d := 5 (2)
```

```
> f1 := Randprime(d,x) mod p;
f1 := x^5 + 3 x^4 + 2 x^3 + 3 x^2 + 9 x + 1 (3)
```

```
> f2 := Randprime(d,x) mod p;
f2 := x^5 + 3 x^4 + 2 x^3 + 6 x^2 + 10 x + 2 (4)
```

```
> a := Expand(f1*f2) mod p;
a := x^10 + 6 x^9 + 2 x^8 + 10 x^7 + 6 x^6 + x^5 + 10 x^4 + 2 x^3 + 3 x^2 + 6 x + 2 (5)
```

```
> Gcd( x^p-x, a ) mod p;
1 (6)
```

```
> st := time();
for k to 5 do
  x^(p^k)-x, GCD = Gcd( a, x^(p^k)-x ) mod p;
od;
time()-st;
st := 60.626
x^11 - x, GCD = 1
x^121 - x, GCD = 1
x^1331 - x, GCD = 1
x^14641 - x, GCD = 1
x^161051 - x, GCD = x^10 + 6 x^9 + 2 x^8 + 10 x^7 + 6 x^6 + x^5 + 10 x^4 + 2 x^3 + 3 x^2 + 6 x + 2
0.028 (7)
```

Use $\gcd(x^{(p^k)} - x, a) = \gcd((x^{(p^k)} \bmod a) - x, a)$

```
> st := time();
w := x;
for k to d do
  w := Rem( w^p, a, x ) mod p;
  print( GCD = Gcd(a,w-x) mod p );
od;
time()-st;
w := x
GCD = 1
GCD = 1
GCD = 1
GCD = 1
GCD = x^10 + 6 x^9 + 2 x^8 + 10 x^7 + 6 x^6 + x^5 + 10 x^4 + 2 x^3 + 3 x^2 + 6 x + 2
0.014 (8)
```

To split use $\gcd\left(v^{\frac{(p^k-1)}{2}} - 1, a\right)$

```
> k := 5;
k := 5 (9)
```

```
> v := Randpoly(k,x) mod p;
```

$$v := 4x^5 + 10x^4 + 4x^3 + 4x^2 + 10x + 10$$
(10)

```
> v^((p^k-1)/2)-1;
```

$$(4x^5 + 10x^4 + 4x^3 + 4x^2 + 10x + 10)^{80525} - 1$$
(11)

```
> w := Powmod( v, (p^k-1)/2, a, x ) mod p;
```

$$w := 6x^9 + 8x^8 + 9x^7 + 6x^6 + 2x^5 + 5x^4 + 3x^3 + 8x^2 + 4x + 1$$
(12)

```
> g := Gcd( a, w-1 ) mod p;
```

$$g := x^5 + 3x^4 + 2x^3 + 3x^2 + 9x + 1$$
(13)

```
> Divide(a,g,'f') mod p;
```

true

(14)

```
> f;
```

$$x^5 + 3x^4 + 2x^3 + 6x^2 + 10x + 2$$
(15)

```
> good := 0:
n := 100:
to n do
  v := Randpoly(k,x) mod p;
  w := Powmod( v, (p^k-1)/2, a, x ) mod p;
  g := Gcd(a,w-1) mod p;
  if degree(g)=k then good += 1; fi;
od:
good;
```

55

(16)